

## POPULATION GROWTH

By David Ottalini  
Washington Apple Pi  
/// SIG Co-Chairman

/// SIG member Tim Boquet submitted a great little program this month for all to enjoy. It gives you a running count of the world's population. To run it, you will need the BYTE font (which is in the Character.Sets subdirectory on side two of /// SIG disk 1004 (System Utilities and Data) as well as the Download and Bgraf invokable modules (they are on many of our PD disks). The program goes like this:

```
10  REM world population program by TIM BOUQUET
20  REM based on facts from the Washington Post
30  REM A) World population reached 5 billion in 1986
40  REM B) World population will be 6 billion by 2000
50  REM
60  REM AUTHOR grants copy permission to Washington Apple Pi
70  REM Author Bouquet retains all other rights 5/22/88
80  REM
90  REM Author wishes to thank WAP for programming ideas!!
100 REM
110 HOME
115 REM Be sure to set your prefix (location of disk with this
116 REM program on it before running. For example: 117 Prefix$=".d1"
120 INVOKE"download.inv"
130 DIM a%(511),a$(20),name$(20)
140 a$="byte":array$="A%"
150 name$=CHR$(34)+a$+CHR$(34)
160 PERFORM getfont(@name$, @array$)
170 PERFORM loadfont(@array$)
180 CLEAR:PRINT
190 DIM A(12),R$(12)
200 OPEN#1,".grafix"
210 INVOKE"bgraf.inv"
220 PERFORM fillport
230 k=0:VPOS=1:HPOS=28
240 PERFORM moveto(%90,%188)
250 PRINT#1;"WORLD POPULATION"
260 PERFORM grafixon
```

```

270  FOR I=1 TO 12:READ A(I):NEXT I
280  FOR I=1 TO 12:READ R$(I):NEXT I
290  ON KBD GOTO 740
300  VPOS=4:HPOS=1
310  yr$=LEFT$(DATE$,2)
320  yr=VAL(yr$)
330  mm$=MID$(DATE$,4,2)
340  mm=VAL(mm$)
350  dd$=RIGHT$(DATE$,2)
360  dd=VAL(dd$)
370  d2$=dd$+r$(mm)+"19"+yr$
380  HPOS=2
390  tttt$= TIME$
400  REM PRINT d2$," ",Tttt$
410  PERFORM moveto(%90,%140)
420  PRINT#1;d2$
430  PERFORM moveto(%120,%120)
440  PRINT#1;tttt$
450  REM DATE of the year - JULIAN
460  jd=0
470  FOR i=1 TO mm:jd=jd+a(i):NEXT i
480  jd=jd+dd
490  REM pop per year day hour min sec
500  ppy=10000000000/14
510  ppd=ppy/365
520  pph=ppd/24
530  ppm=pph/60
540  pps=ppm/60
550  REM time of day
560  hh$=LEFT$(Tttt$,2)
570  hh=VAL(hh$)
580  mm$=MID$(Tttt$,4,2)
590  mm=VAL(mm$)
600  ss$=RIGHT$(Tttt$,2)
610  ss=VAL(ss$)
620  REM current population
630  pd&=CONV&(jd*ppd):ph&=CONV&(pph*hh)
640  pm&=CONV&(ppm*mm):ps&=CONV&(pps*ss)
650  dhm&=pm&+ph&+pd&
660  py&=CONV&(ppy*(yr-86.0))+CONV&(5.0E+9)
670  ppp&=dhm&+py&
680  HPOS=30

```

```

690 REM PRINT ppp&+ps&
700 PERFORM moveto(%100,%170)
710 PRINT#1 USING"13&";ppp&+ps&
720 IF VPOS>20 THEN VPOS=4
730 GOTO 310
740 REM: RUN"hello" here if you wish.
750 END
760 DATA 31,28,31,30,31,30,31,31,30,31,30,31
770 DATA " ** January ** "," ** february ** "," ** MARCH ** "," ** APRIL **"
780 DATA " ** MAY ** "," ** June ** "," ** JULY ** "," ** AUGUST ** "
790 DATA " ** SEPTEMBER ** "," ** OCTOBER ** "," ** NOVEMBER ** "," ** DECEMBE
R* "

```

Note that once you've run this program, you'll be stuck with the BYTE character set while in Basic (or you could use the Newdownload font utility on disk 1002 to change the font). Booting a new program, by the way, will usually load the Standard Character set.

Tim also had a few wishes for SIG disks including "a version of the menu.maker which will run pascal programs" (Co-Chairman Tom Bartkiewicz has a Pascal program in Beta form now to do that, but you can't run Pascal programs from Basic or Basic from Pascal). Tim would also like to see Menu.Maker be able to display foto files, and change fonts when font files are highlighted. Both are possible, but add to the overhead for the program, since it takes more invokables and disk space (which for SIG disks means less program space).

The neat thing about Menu.Maker however is its adaptability. It would not be that difficult to let it recognize a foto file and display it on the screen (our "Run.Side.Two" program on disk 1012 does that). And you could adapt the Newdownload program to run from within Menu.Maker to change fonts depending on which one was displayed (in fact, you could adapt the font download routine in the above program). By the way, remember we are now up to the 3.0 version of Menu.Maker, which is available on the latest PD disks. We are working to upgrade older disks as well.

## FINALLY

Due to recent demand for our "How to Use the Apple ///" video tape, we have added a second copy in the WAP office. Remember there is a \$20.00 deposit required before you can take out the tape.

The summer is once again upon us, and that prompts a reminder that there will be NO August SIG meeting (there will be a Trail article, though). Attendance has been poor of late and hopefully a little breather will be good for the /// soul. Until September,

then, have a safe and fun summer!